



THE AMSTRAD PENTACLE ADVENTURE CREATOR

Clive Gifford



Just £2.99!
Just type in the program, and use it to create an infinite number of exciting adventures on your Amstrad. The PENTACLE program does all the hard work. As well as PENTACLE, this publication includes full data for an adventure set in medieval times, Castle of Doom, and a full graphical adventure, The Search for SPECK. You can sell programs created with PENTACLE, without restriction!

Interface Publications
LONDON · MELBOURNE

**THE
AMSTRAD
PENTACLE
ADVENTURE
CREATOR**

Clive Gifford



THE AMSTRAD PENTACLE

CONTENTS

| | |
|--|----|
| Foreword – Tim Hartnell | 5 |
| PENTACLE program listing | 7 |
| PENTACLE instructions | 31 |
| CASTLE OF DOOM (medieval) | 35 |
| THE SEARCH FOR SPECK (graphical) | 41 |
| LEAKED! (political) | 51 |
| Creating your Own | 59 |
| 50 Dynamic Games | 63 |

Acknowledgements and Dedication

To Catherine

The author wishes to thank a number of people who, in their own way have helped with the writing of this book. To Ian for unknowingly keeping me sane while I battled with Pentacle. To Richard, Jeff, Caroline E. and C.Y. for their encouraging letters. To Brian for supplying an excellent adventure. Last, but not least, to my family for tolerance and caring.

First published in the UK by:
Interface Publications Ltd.,
9-11 Kensington High Street
LONDON W8 5NP

Copyright © Tim Hartnell, 1985
First printing May 1985

ISBN 0 907563 79 8

The programs in this book have been included for their instructional value. They have been tested with care, but are not guaranteed for any particular purpose. While every care has been taken, the publishers cannot be held responsible for any running mistakes which may occur.

ALL RIGHTS RESERVED

No use whatsoever may be made of the contents of this volume – programs and/or text – except for private study by the purchaser of this volume, without the prior written permission of the copyright holder.

Reproduction in any form or for any purpose is forbidden.

Books published by Interface Publications are distributed in the UK by WHS Distributors, St. John's House, East Street, Leicester LE1 6NE (0533 551196) and in Australia and New Zealand by PITMAN PUBLISHING. Any queries regarding the contents of this volume should be directed by mail to Interface Publications, 9-11 Kensington High Street, London W8 5NP.

Foreword **TIM HARTNELL**

Clive Gifford is an adventure nut. From the early, crude ZX81 adventures which came on the market, to the highly-sophisticated Level 9 and Scott Adams' software you can get today for a host of machines, Clive has played his way through a hundred or more adventures, gaining considerable skill in the process.

It wasn't just all play. Along the way, Clive managed to write a number of adventure books, which showed his growing prowess in the genre. From the Interface Publications' books *Creating Adventures on your Dragon 32* and *Creating Adventures on your Commodore 64*, through to the Virgin Books' *Adventures for your Spectrum* and *More Adventures for your Spectrum*, Clive has been developing and extending his adventure-writing skills.

In this book, easily his strongest work to date, Clive shares with you the exciting program *Pentacle* which actually allows you to create and run your own adventure programs on the Amstrad, without having to go to the trouble of actually writing the program yourself. Think up the locations and the story-line, and *Pentacle* takes over all the boring bits of writing the program.

He then shows you *Pentacle* in action to create an adventure set in medieval times, *Castle of Doom*.

Then, the graphical potential of the Amstrad gets a real workout with *The Search for SPECK*.

Time to go adventuring . . .

**Tim Hartnell,
London, April 1985**

Tim Hartnell is one of the most widely-published computer authors in the world. His books include Games BBC Computers Play (Addison-Wesley, 1982); Tim Hartnell's QL Handbook (Interface Publications Ltd., 1984); Creating Adventure Programs on your Microcomputer (Ballantine, US, 1984); and The Personal Computer Handbook (Virgin Books, 1984). He is managing director of the specialist publishing house Interface Publications Ltd., and writes the regular column Response Frame for Your Computer magazine, as well as contributing to ZX Computing, Your Spectrum and Microscope.

Pentacle

Pentacle is an adventure creator that allows you to create complex adventures of up to 120 locations with 150 objects, 40 monsters, 40 readable messages and a whole host of tricks and puzzles which the adventure player must solve. The great thing about *Pentacle* is that you need little or no programming knowledge or particular skill to create your adventure.

How does *Pentacle* work? You simply type in your adventure information, as prompted by the computer. Your Amstrad stores this information, placing it into an adventure framework, thus creating the adventure program.

The whole idea, as I'm sure you can appreciate, is a complicated one to program. The greatest problem is allowing for as much flexibility as possible, so that you can create an adventure that is pretty original. Working against flexibility is the need for a fully running, user-friendly program which allows the data to be typed in and easily stored and retrieved. All of this had to be achieved in BASIC and the whole program should allow the creation of both tiny, simple adventures as well as complex monsters that could take ages to solve.

I have included the data to play a small adventure in the standard, medieval mould, entitled, *Castle of Doom*.

```
9 REM*****
10 REM*****PENTACLE*****
11 REM*****
20 CLEAR
30 GOSUB 3880
40 PH=18:GOSUB 3940
50 REM*****MENU*****
60 BORDER 1:INK 1,24:INK 0,1
70 CLS
80 MODE 1
90 PRINT " A D V E N T U R E      C R E
  A T O R"
100 PRINT "=====
====="
```


Amstrad PENTACLE

```
110 PRINT :PRINT "      INPUT DATA
      PRESS 1"
120 PRINT :PRINT "      PLAY ADVENTURE
      PRESS 2"
130 PRINT :PRINT "      SAVE DATA
      PRESS 3"
140 PRINT :PRINT "      LOAD DATA
      PRESS 4"
150 PRINT :PRINT "      PRINT DATA
      PRESS 5"
160 PRINT: PRINT "      MEMORISE DATA
      PRESS 6"
170 FOR T=1 TO 22000:A$=INKEY$:IF A$<>" "
      THEN T=22000
180 NEXT
190 IF VAL(A$)>7 OR VAL(A$)<1 THEN GOTO
170
200 SOUND 1,180,100
210 ON VAL(A$) GOTO 220,1220,3190,3390,4
140,4570
220 REM***INPUT DATA***
230 IF OB(1,1)<>0 THEN GOSUB 3880
240 CLS
250 PRINT TAB(7);"LOCATION DESIGN"
260 PRINT TAB(7);"===== "
270 PRINT
280 INPUT "PLEASE TYPE IN THE ADVENTURE
TITLE";MG$(3)
290 CLS
300 INPUT "HOW MANY LOCATIONS (max 120)"
;N
310 IF N>120 OR N<1 THEN 300
```



```
320 FOR T=1 TO N
330 PRINT:PRINT
340 PRINT "ENTER NAME/DESCRIPTION OF LOC
ATION";T
350 INPUT "          ";L$(T)
360 NEXT
370 PRINT:PRINT:INPUT "TO GO BACK TO THE
MENU ENTER 999";A
380 IF A=999 THEN GOTO 40
390 FOR T=1 TO N
400 CLS
410 PRINT:PRINT "          ";L$(T)
420 IF LEN(L$(T))>32 THEN PRINT LEFT$(M$
,32):GOTO 440
430 PRINT "          ";LEFT$(M$,LEN(L$(T))
)
440 FOR X=1 TO 6
450 PRINT "IF YOU GO ";MID$(DD$, (X*5)-4,
5):INPUT"WHICH LOCATION WILL YOU REACH";
L(T,X)
460 NEXT
470 PRINT
480 INPUT "IF THE PLAYER USES A MAGIC CO
MMAND HERE WHERE WILL HE END UP";L(T,7)
490 PRINT
500 INPUT "IF THE PLAYER RUNS OR JUMPS A
T THIS          LOCATION, WHERE WILL HE END U
P";L(T,8)
510 NEXT T
520 REM*****OBJECTS*****
530 CLS
```

```
540 PRINT TAB(16); "OBJECTS"
550 PRINT TAB(16); "======"
560 PRINT
570 INPUT "HOW MANY OBJECTS IN THIS ADVE
NTURE (max 150) "; OM
580 IF OM>150 THEN 570
590 FOR T=1 TO OM
600 CLS
610 PRINT:INPUT "NAME OF OBJECT"; OB*(T)
620 PRINT:INPUT "THE OBJECT'S ORIGINAL L
OCATION"; OB(T,1)
630 PRINT:INPUT "THE OBJECT'S WEIGHT (in
roughly pound units)"; OB(T,2)
640 INPUT "IS THE OBJECT IMPOSSIBLE TO D
ROP ONCE TAKEN (1), ONLY POSSIBLE TO F
IND USING SEARCH/EXAMINE (2) (0= NEITHE
R OF THESE) "; OB(T,3)
650 PRINT:INPUT "IS THE OBJECT; A WEAPON
(2); A TREASURE (3); AN UNLOCKING DEVIC
E (4) (0=NONE OF THESE) "; OB(T,4)
660 NEXT
670 REM***MONSTER GENERATION***
680 CLS
690 PRINT TAB(7); "MONSTER GENERATION"
700 PRINT TAB(7); "======"
710 PRINT
720 INPUT "HOW MANY 'BADDIES' IN YOUR AD
VENTURE (max 40) "; BM
730 IF BM>40 THEN 720
740 FOR T=1 TO BM
750 PRINT:PRINT "NAME OF ENEMY, FOLLOWED
```



```
BY ITS POSITION AND ITS STRENGTH (1-50)
"
760 INPUT MO$(T)
770 INPUT MO(T,1):INPUT " ";MO(T,2)
780 NEXT
790 REM***LOCKED/UNLOCKED DOORS***
800 CLS
810 PRINT TAB(7);"OPEN AND CLOSED DOORS"
820 PRINT TAB(7);"=====
830 PRINT,,,"TYPE 999 FOR NO DOORS"
840 PRINT:PRINT
850 INPUT "LOCKED DOOR BETWEEN ";L
860 IF L=999 THEN 970
870 INPUT "AND";LL
880 PRINT L$(L)
890 INPUT "WHICH DIRECTION IS THE LOCKED
DOOR (N=1,S=2,E=3,W=4,U=5,D=6)";D
900 IF D>6 OR D<1 THEN 890
910 PRINT:PRINT L$(LL)
920 INPUT "WHICH DIRECTION IS THE LOCKED
DOOR (N=1 etc.)";DD
930 IF DD>6 OR DD<1 THEN 920
940 L(L,D)=-LL:L(LL,DD)=-L
950 PRINT:PRINT:INPUT "DO YOU WANT ANY M
ORE DOORS (Y/N)";A$
960 IF LEFT$(A$,1)="Y" THEN GOTO 840
970 REM*****GAME RULES*****
980 CLS
990 PRINT TAB(8);"GAME RULES AND PUZZLES
"
1000 PRINT TAB(8);"=====
="
```

```
1010 PRINT:PRINT
1020 INPUT "ENTER THE STARTING STRENGTH
(WHICH IS ONLY USED IN COMBAT)";ST
1030 PRINT:IF ST<1 THEN ST=10000
1040 CS=ST
1050 INPUT "MAXIMUM WEIGHT PLAYER CAN CA
RRY";MW
1060 IF MW<1 THEN MW=1000
1070 PRINT:INPUT "ENTER MESSAGE WHEN PLA
YER ASKS FOR HELP";MG$(1)
1080 PRINT:INPUT "THE NUMBER OBJECT REQU
IRED TO ALLOW THE PLAYER TO SEARCH/EXAMI
NE (0=NO OBJECT NECESSARY) ";SO
1090 PRINT:INPUT "MAXIMUM NUMBER OF MOVE
S";MM:IF MM<1 THEN MM=10000
1100 PRINT:INPUT "ENTER THE MAGIC COMMAN
D";MG$(2)

1120 PRINT:INPUT "HOW MANY TREASURES DO
YOU NEED BEFORE YOU HAVE WON ";WT
1130 PRINT:INPUT "HOW MANY MONSTERS MUST
YOU HAVE KILLED BEFORE YOU HAVE WON ";
WM
1140 PRINT:INPUT "WHICH NUMBER LOCATION
MUST YOU BE AT BEFORE YOU HAVE WON";W
L
1150 PRINT:INPUT "DO YOU WANT THE PLAYER
TO KNOW OF THE EXITS (1=Yes, 0=No) ";
E
1160 T=1
1170 PRINT:INPUT "ENTER READABLE MESSAGE
```

```
FOLLOWED BY NO. OF ITS LOCATION (ENTER
'999' IF NO MESSAGE) ";A$
1180 IF A$="999" THEN RM=T:GOTO 40
1190 R$(T)=A$:INPUT R(T)
1200 T=T+1:GOTO 1170
1210 GOTO 40
1220 REM****PLAY ADVENTURE****
1230 ST=CS:P=1:MV=1

1240 REM****MAIN ADVENTURE GAME****
1250 CLS
1260 S=INT((40-LEN(MG$(3)))/2)
1270 PRINT TAB(S);MG$(3)
1280 PRINT:PRINT
1290 PRINT "YOU ARE ";L$(P)
1300 PRINT:PRINT
1310 IF E<>1 THEN GOTO 1370
1320 PRINT "VISIBLE EXITS ARE ";
1330 FOR T=1 TO 6:IF L(P,T)>0 THEN PRINT
MID$(DD$(T*5)-4,5);

1340 IF T<3 THEN PRINT " ";
1350 NEXT T
1360 PRINT:PRINT
1370 PRINT "YOU CAN SEE:";
1380 FOR T=1 TO OM
1390 IF OB(T,1)=P AND OB(T,3)<>2 THEN PR
INT TAB(14);OB$(T)
1400 NEXT T
1410 PRINT:PRINT
1420 REM****HAZARDS****
1430 IF MV>MM THEN PRINT "YOU TOOK TOO L
ONG BUDDY":FOR T=1 TO 2000:NEXT:GOTO 381
0
```

```
1440 IF (L(P,1)=0 AND L(P,2)=0 AND L(P,3
)=0 AND L(P,4)=0 AND L(P,5)=0 AND L(P,6)
=0 AND L(P,8)=0) AND (L(P,7)=0 OR (L(P,7
)>0 AND RND>0.8)) THEN 3810
1450 MF=0:FOR T=1 TO BM
1460 IF MO(T,1)=P THEN PRINT MO$(T);" AT
TACKS...":MF=1:FM=T:T=BM
1470 NEXT

1480 MT=0:FOR T=1 TO OM
1490 IF OB(T,1)=P OR OB(T,1)=0 AND OB(T,
4)=3 THEN MT=MT+1
1500 NEXT
1510 FOR T=1 TO BM
1520 IF MO(T,1)=-1 THEN MT=MT+1
1530 NEXT
1540 IF MT>=(WT+WM) AND P=WL THEN GOTO 3
710
1550 E$="":C$="":N$=""

1560 LOCATE 1,15:INPUT " WHAT NOW ";C$:E
$=C$
1570 PRINT:PRINT:MV=MV+1
1580 FOR T=1 TO LEN(E$):IF MID$(E$,T,1)=
" " THEN C$=LEFT$(E$,T-1):N$=RIGHT$(E$, (
LEN(E$)-T)):T=LEN(E$)
1590 NEXT:CV=0
1600 IF C$=MG$(2) THEN 3050
1610 FOR T=1 TO 35
1620 IF C$=F$(T) THEN CV=T:T=35
1630 NEXT
1640 IF MF=1 AND (CV<17 OR CV>20) AND CV
```



```
<>34 AND CV<>35 THEN PRINT "NO YOU MUST
STAY AND FIGHT":ST=ST-5:GOTO 1790
1650 IF CV=35 THEN GOTO 40
1660 IF CV>12 THEN GOTO 1780
1670 IF CV=0 THEN PRINT "I CAN'T ";C$;"
";N$:GOTO 1790
1680 PRINT:PRINT
1690 SOUND 1,180:PP=0

1700 IF LEFT$(C$,1)="N" AND L(P,1)>0 THE
N P=L(P,1):PP=1
1710 IF LEFT$(C$,1)="S" AND L(P,2)>0 THE
N P=L(P,2):PP=1
1720 IF LEFT$(C$,1)="E" AND L(P,3)>0 THE
N P=L(P,3):PP=1
1730 IF LEFT$(C$,1)="W" AND L(P,4)>0 THE
N P=L(P,4):PP=1
1740 IF LEFT$(C$,1)="U" AND L(P,5)>0 THE
N P=L(P,5):PP=1
1750 IF LEFT$(C$,1)="D" AND L(P,6)>0 THE
N P=L(P,6):PP=1

1760 IF PP=1 THEN PP=0:PRINT " MOVING NO
W.....":FOR T=200 TO 60 STEP -20:SOUN
D 1,T:NEXT T:GOTO 1240
1770 PRINT "CANNOT MOVE THAT WAY":GOTO 1
790
1780 ON CV-12 GOSUB 1810,1810,1920,1920,
2000,2000,2000,2000,2090,2230,2370,2430,
2490,2560,2560,2630,2630,2700,2700,2810,
2960,2960
1790 SOUND 1,180:WHILE INKEY$="":WEND
1800 GOTO 1240
```

```
1810 REM*****TAKE/GET*****
1820 OV=0
1830 FOR T=1 TO OM: IF OB$(T)=N$ THEN OV=
T:T=OM
1840 NEXT
1850 IF OV=0 THEN PRINT "NO SUCH OBJECT
HERE I'M AFRAID":RETURN
1860 IF OB(OV,1)=0 THEN PRINT "YOU ALREA
DY HAVE IT!":RETURN
1870 IF OB(OV,2)>MW THEN PRINT "IMPOSSIB
LE TO CARRY":GOTO 1890
1880 IF OB(OV,2)>(MW-WC) THEN PRINT "TOO
HEAVY, AT THE MOMENT":RETURN
1890 IF OB(OV,1)=P AND OB(OV,2)<(MW-WC)
THEN OB(OV,1)=0:WC=WC+OB(OV,2):PRINT "O.
K. THAT'S TAKEN"
1900 IF OB(OV,4)=1 THEN KC=1
1910 RETURN
1920 REM*****DROP/LEAVE*****
1930 OV=0
1940 FOR T=1 TO OM: IF OB$(T)=N$ THEN OV=
T:T=OM
1950 NEXT
1960 IF OV=0 THEN PRINT "NO SUCH OBJECT
TO LEAVE, I'M AFRAID":RETURN
1970 IF OB(OV,3)=1 THEN PRINT "YOU CANNO
T DROP IT":RETURN
1980 IF OB(OV,1)<>0 THEN PRINT "YOU ARE
NOT CARRYING THAT OBJECT":RETURN
1990 OB(OV,1)=P:PRINT "YOU'VE DROPPED ":
PRINT OB$(OV):WC=WC-OB(OV,2):RETURN
```

```
2000 REM***FIGHT/KILL/ATTACK/HIT***
2010 CLS:PRINT:PRINT "    LET THE BATTLE
    COMMENCE...."
2020 SOUND 1,200,60:SOUND 1,400,60:SOUND
    1,300,60
2030 FOR T=1 TO 0M:IF OB(T,1)=0 AND OB(T
    ,4)=2 THEN MO(FM,2)=INT((MO(FM,2))/2)
2040 NEXT
2050 ST=ST-MO(FM,2)
2060 MO(FM,1)=-1
2070 IF ST<1 THEN PRINT "YOU ARE EXHAUST
    ED...YOU HAVE BEEN          DEFEATED IN BAT
    TLE":FOR T=1 TO 2000:NEXT:GOTO 3810
```



```
2080 PRINT:PRINT "          YOU ARE THE V
ICTOR":RETURN
2090 REM*****UNLOCK*****
2100 K=0
2110 FOR T=1 TO OM:IF OB(T,1)=0 AND OB(T
,4)=4 THEN K=1
2120 NEXT:LD=0
2130 IF K<>1 THEN PRINT "YOU HAVE NO UNL
OCKING DEVICE":RETURN
2140 IF LEFT$(N$,1)="N" AND L(P,1)<0 THE
N L(P,1)=ABS(L(P,1)):LD=1
2150 IF LEFT$(N$,2)="S" AND L(P,2)<0 THE
N L(P,2)=ABS(L(P,2)):LD=1
2160 IF LEFT$(N$,3)="E" AND L(P,3)<0 THE
N L(P,3)=ABS(L(P,3)):LD=1
2170 IF LEFT$(N$,4)="W" AND L(P,4)<0 THE
N L(P,4)=ABS(L(P,4)):LD=1
2180 IF LEFT$(N$,5)="U" AND L(P,5)<0 THE
N L(P,5)=ABS(L(P,5)):LD=1
2190 IF LEFT$(N$,6)="D" AND L(P,6)<0 THE
N L(P,6)=ABS(L(P,6)):LD=1
2200 IF LD=1 THEN PRINT "THE DOOR SLOWLY
OPENS..."
2210 IF LD=1 THEN FOR T=850 TO 300 STEP-
10:SOUND 1,T,2:SOUND 1,0,2:NEXT T:RETURN
2220 PRINT "THERE'S NO LOCKED DOOR NEARB
Y":RETURN
2230 REM*****LOCK*****
2240 K=0
2250 FOR T=1 TO OM:IF OB(T,1)=0 AND OB(T
,4)=4 THEN K=1
2260 NEXT:LD=0
```



```
2270 IF K<>1 THEN PRINT "YOU HAVE NO LOC
KING DEVICE":RETURN
2280 IF LEFT$(N$,1)="N" AND L(P,1)>0 THE
N L(P,1)=-L(P,1):LD=1
2290 IF LEFT$(N$,2)="S" AND L(P,2)>0 THE
N L(P,2)=-L(P,2):LD=1
2300 IF LEFT$(N$,3)="E" AND L(P,3)>0 THE
N L(P,3)=-L(P,3):LD=1
2310 IF LEFT$(N$,4)="W" AND L(P,4)>0 THE
N L(P,4)=-L(P,4):LD=1
2320 IF LEFT$(N$,5)="U" AND L(P,5)>0 THE
N L(P,5)=-L(P,5):LD=1
2330 IF LEFT$(N$,6)="D" AND L(P,6)>0 THE
N L(P,6)=-L(P,6):LD=1
2340 IF LD=1 THEN PRINT "THE DOOR SLOWLY
CLOSES..."
2350 IF LD=1 THEN FOR T=300 TO 850 STEP
10: SOUND 1,T,2: SOUND 1,0,2: NEXT T: RETURN
2360 PRINT "THERE'S NO UNLOCKED DOOR NEA
RBY": RETURN
2370 REM*****WAIT*****
2380 LOCATE 1,18
2390 PRINT "WAITING"
2400 FOR T=1 TO 5000: NEXT
2410 MV=MV+5
2420 RETURN
2430 REM*****HELP*****
2440 PEN 2
2450 PRINT MG$(1)
2460 SOUND 1,400,60
2470 PEN 1
2480 RETURN
```

```

2490 REM*****READ*****
2500 JJ=0
2510 FOR T=1 TO 40
2520 IF P=R(T) THEN PRINT TAB(10);"MESSA
GE READS...":PRINT:PRINT R$(T):T=40:JJ=1
2530 NEXT
2540 IF JJ<>1 THEN PRINT "THERE IS NOTHI
NG TO READ"
2550 RETURN
2560 REM*****SEARCH/EXAMINE*****
2570 IF S0=0 THEN 2590
2580 IF OB(S0,1)<>0 THEN PRINT "YOU NEED
SOMETHING ELSE":RETURN
2590 FOR T=1 TO OM:IF OB(T,1)=P AND OB(T
,3)=2 THEN OB(T,3)=0:PRINT "YOU CAN SEE
SOMETHING NEW...":PX=1:T=OM
2600 NEXT
2610 IF PX=1 THEN PX=0 ELSE PRINT "YOU H
AVE FOUND NOTHING IN YOUR LONG SEARCH
H":MV=MV+5
2620 RETURN
2630 REM*****QUIT/FINISH*****
2640 PRINT " DO YOU REALLY WANT TO QUI
T?"
2650 PRINT " ====="
2660 PRINT:INPUT "YES OR NO";A$
2670 IF LEFT$(A$,1)="Y" THEN INK 0,0:INK
1,1:PRINT:FOR T=20 TO 500:INK 1,INT(T/2
0):PRINT "COWARD!!":NEXT:GOTO 40
2680 PRINT "I THOUGHT YOU WANTED TO CARR
Y ON THAT'S WHY I GAVE YOU ANOTHER CHANC
E"

```



```
2950 FOR T=1 TO 2000:NEXT:RETURN
2960 REM*****JUMP*****
2970 IF L(P,8)=0 AND C$="RUN" THEN PRINT
  "NOWHERE TO RUN TO":RETURN
2980 IF L(P,8)>0 AND C$="RUN" THEN PRINT
  "RUNNING NOW":ST=ST-5: MV=MV+8:RETURN
2990 IF L(P,8)=0 THEN PRINT "THERE IS NO
WHERE TO JUMP TO":RETURN
3000 IF L(P,8)>0 THEN P=L(P,8):PRINT "JU
MPING NOW"
3010 FOR SD=140 TO 60 STEP-20:SOUND 1,SD
,20,15:NEXT
3020 FOR SD=60 TO 140 STEP 20:SOUND 1,SD
,20,15:NEXT
3030 ST=ST-5: MV=MV+10
3040 RETURN
3050 REM*****MAGIC COMMAND*****
3060 IF L(P,7)<1 OR L(P,7)=P THEN PRINT
"THE COMMAND HAS NO EFFECT, YOU HAVE
WASTED TIME HERE":GOTO 1790
3070 BORDER 24,6
3080 CLS
3090 INK 3,2,7:PEN 3:INK 0,26
3100 P=L(P,7)
3110 FOR T=180 TO 60 STEP-2:SOUND 1,T,1,
15:NEXT
3120 LOCATE 11,10:PRINT "THE MAGIC WORKS
..."
3130 LOCATE 6,12:PRINT "YOU ARE MYSTICAL
LY TRANSPORTED"
3140 LOCATE 12,13:PRINT "TO ANOTHER PLAC
E"
```

```
3150 FOR T=1 TO 1000:NEXT T
3160 PEN 1:BORDER 1:INK 0,1
3170 GOTO 1790
3180 REM*****
3190 REM***SAVE DATA***
3200 CLS:LOCATE 4,10
3210 INPUT "FILE NAME ";Y$
3220 PEN 3
3230 ERASE L$,L,OB$,OB,MO$,MO,R$,R
3240 PRINT:PRINT:PRINT TAB(5);
3250 OPENOUT Y$
3260 PRINT #9,N,OM,BM,RM
3270 FOR T=1 TO N:PRINT #9,K$(T):NEXT T
3280 FOR T=1 TO N:FOR X=1 TO 8:PRINT #9,
K(T,X):NEXT:NEXT
3290 FOR T=1 TO OM:PRINT #9,NA$(T):NEXT
3300 FOR T=1 TO OM:FOR X=1 TO 4:PRINT #9
,NA(T,X):NEXT:NEXT
3310 FOR T=1 TO BM:PRINT #9,LN$(T):NEXT
3320 FOR T=1 TO BM:PRINT #9,LN(T,1):PRIN
T #9,LN(T,2):NEXT
3330 FOR T=1 TO RM:PRINT #9,Q$(T):NEXT
3340 FOR T=1 TO RM:PRINT #9,Q(T):NEXT
3350 PRINT #9,CS;MW;SO;MM;WL;WT;WM;E
3360 FOR T=1 TO 3:PRINT #9,MG$(T):NEXT
3370 CLOSEOUT
3380 PEN 1:GOTO 40
3390 REM***LOAD DATA***
3400 CLS:LOCATE 1,11
3410 INPUT "HAVE THE USELESS ARRAYS BEEN
ERASED? Y/N";A$
```

```
3420 IF LEFT$(A$,1)="Y" THEN 3440
3430 ERASE L$,L,OB$,OB,MO$,MO,R$,R
3440 X$=""
3450 OPENIN X$
3460 INPUT #9,N,OM,BM,RM
3470 FOR T=1 TO N: INPUT #9,K$(T):NEXT
3480 FOR T=1 TO N:FOR X=1 TO B: INPUT #9,
K(T,X):NEXT:NEXT
3490 FOR T=1 TO OM: INPUT #9,NA$(T):NEXT
3500 FOR T=1 TO OM:FOR X=1 TO 4: INPUT #9
,NA(T,X):NEXT:NEXT
3510 FOR T=1 TO BM: INPUT #9,LN$(T):NEXT
3520 FOR T=1 TO BM: INPUT #9,LN(T,1): INPU
T #9,LN(T,2):NEXT
3530 FOR T=1 TO RM: INPUT #9,Q$(T):NEXT
3540 FOR T=1 TO RM: INPUT #9,Q(T):NEXT
3550 INPUT #9,CS:GOSUB 3690
3560 INPUT #9,MW:GOSUB 3690
3570 INPUT #9,SD:GOSUB 3690
3580 INPUT #9,MM:GOSUB 3690
3590 INPUT #9,WL:GOSUB 3690
3600 INPUT #9,WT:GOSUB 3690
3610 INPUT #9,WM:GOSUB 3690
3620 INPUT #9,E:GOSUB 3690
3630 FOR T=1 TO 3: INPUT #9,MG$(T):NEXT
3660 CLOSEIN
3670 PH=0:GOSUB 3910
3680 GOTO 40
3690 IF EOF=-1 THEN CLOSEIN:GOTO 3670
3700 RETURN
3710 REM***** PLAYER WINS *****
3720 CLS
```



```
3730 LOCATE 14,8
3740 PRINT "WELL DONE!!!"
3750 LOCATE 13,12
3760 PRINT "YOU'VE MADE IT"
3770 LOCATE 7,16
3780 PRINT "YOU'VE SOLVED THIS ADVENTURE
"
3790 BORDER 3,12
3800 WHILE INKEY$="":WEND:GOTO 40
3810 REM***** PLAYER LOSES *****
3820 FOR T=1 TO 2500:NEXT T
3830 CLS
3840 FOR T=1 TO 10:PRINT:NEXT T
3850 PRINT "      YOU FAILED OH PUNY ONE
."
3860 BORDER 0:SOUND 1,3000,200,15
3870 WHILE INKEY$="":WEND:GOTO 40
3880 REM*****INITIALISATION*****
3890 DIM K$(120),K(120,8),NA$(150),NA(15
0,4),LN$(40),LN(40,2),Q$(40),Q(40)
3900 DIM F$(35)
3910 DIM L$(120),L(120,8),OB$(150),OB(15
0,4),MO$(40),MO(40,2),R$(40),R(40)
3920 M$="-----
-----"
3930 INK 1,24:PEN 1
3940 DD$="NORTHSOUTHEAST WEST UP   DOWN
"
3950 RESTORE
3960 FOR T=1 TO 35
3970 READ F$(T)
3980 NEXT T
```

```
3990 IF PH=18 THEN PH=0:RETURN
4000 FOR T=1 TO N:L$(T)=K$(T):FOR X=1 TO
  8:L(T,X)=K(T,X):NEXT:NEXT
4010 FOR T=1 TO OM:OB$(T)=NA$(T):FOR X=1
  TO 4:OB(T,X)=NA(T,X):NEXT:NEXT
4020 FOR T=1 TO BM:MO$(T)=LN$(T):MO(T,1)
  =LN(T,1):MO(T,2)=LN(T,2):NEXT
4030 T=1
4040 WHILE Q(T)>0
4050 R(T)=Q(T):R$(T)=Q$(T)
4060 T=T+1
4070 WEND
4080 RETURN
4090 DATA NORTH,N,SOUTH,S,EAST,E,WEST,W,
  UP,U,DOWN,D
4100 DATA TAKE,GET,DROP,LEAVE,FIGHT,ATTA
  CK,KILL,HIT
4110 DATA UNLOCK,LOCK,WAIT,HELP,READ
4120 DATA EXAMINE,SEARCH,QUIT,FINISH
4130 DATA LIST,INVENTORY,SCORE,JUMP,RUN,
  MENU
4140 REM*****PRINTOUT DATA*****
4150 CLS
4160 PRINT TAB(15);"PRINT DATA"
4170 PRINT TAB(15);"===== "
4180 LOCATE 12,12
4190 INK 1,1,24:PRINT "PRINTING NOW"
4200 GOSUB 4540
4210 PRINT #8,TAB(10);"ADVENTURE DATA"
4220 GOSUB 4540
4230 PRINT #8,MB$(3)
4240 GOSUB 4540
```



```
4250 FOR T=1 TO N:PRINT #8,L$(T):NEXT
4260 GOSUB 4540
4270 FOR T=1 TO N:PRINT #8,T;" ";;FOR X
=1 TO 8:PRINT #8,L(T,X);:NEXT:PRINT #8,"
":NEXT
4280 GOSUB 4540
4290 FOR T=1 TO OM:PRINT #8,OB$(T):NEXT
T
4300 GOSUB 4540
4310 FOR T=1 TO OM:FOR X=1 TO 4:PRINT #8
,OB(T,X);" ";;NEXT:PRINT #8," "":NEXT
4320 GOSUB 4540
4330 FOR T=1 TO BM:PRINT #8,MO$(T);" ";
MO(T,1);MO(T,2):NEXT
4340 GOSUB 4540
4350 T=1
4360 WHILE R(T)>0
4370 PRINT #8,R$(T);" ";R(T)
4380 T=T+1
4390 WEND
4400 GOSUB 4540
4410 PRINT #8,"COMBAT STRENGTH ";CS
4420 PRINT #8,"MAXIMUM CARRIED WEIGHT ";
MW
4430 PRINT #8,"HELP REPLY :- ";MG$(1)
4440 PRINT #8,"NUMBER OBJECT NEEDED FOR
SEARCH ";SO
4450 PRINT #8,"MAXIMUM NUMBER OF PLAYER
MOVES ";MM
4460 PRINT #8,"MAGIC COMMAND: ";MG$(2)
4470 INK 1,24
4480 PRINT #8,"WINNING LOCATION ";WL
```

```
4490 PRINT #8,"TREASURES NEEDED ";WT
4500 PRINT #8,"MONSTERS SLAIN ";WM
4510 PRINT #8," "
4520 IF E=1 THEN PRINT #8,"PLAYER SEES E
XITS" ELSE PRINT #8,"PLAYER DOESN'T SEE
EXITS"
4530 GOTO 40
4540 REM*****PRINTER BLANKS*****
4550 PRINT #8," ":PRINT #8," "
4560 RETURN
4570 REM*****MEMORISE DATA*****
4580 REM**(TRANSFER DATA TO SPARE ARRAYS
)**
4590 FOR T=1 TO N:K$(T)=L$(T):FOR X=1 TO
8:K(T,X)=L(T,X):NEXT:NEXT
4600 FOR T=1 TO OM:NA$(T)=OB$(T):FOR X=1
TO 4:NA(T,X)=OB(T,X):NEXT:NEXT
4610 FOR T=1 TO BM:LN$(T)=MO$(T):LN(T,1)
=MO(T,1):LN(T,2)=MO(T,2):NEXT
4620 T=1
4630 WHILE R(T)>0
4640 Q(T)=R(T):Q$(T)=R$(T)
4650 T=T+1:WEND
4660 CLS:LOCATE 18,12:PRINT "O.K."
4670 PH=18
4680 GOTO 40
```

Instructions for Pentacle

- (1) Start the program by running it. If, at any time, you have an error or break the program, you can usually continue without any loss of data by typing GOTO 40.
- (2) The menu appears. The usual procedure for creating an adventure is (1) Input Data; (6) Memorise Data; (2) Play Adventure (to check for errors); (3) Save Data. Whenever you are entering data, once you have finished and have returned to the menu, *always* immediately choose the sixth option which allows the data to be transferred to a separate set of arrays.
- (3) Inputting data is going to be the first option you use. This section has been designed to be simple to use. It is worth writing down all the data beforehand, planning it out and then typing it into the computer. To prepare for 'yes/no' questions, enter 0 for no, and the required number for the other answers.
- (4) In the game rules data section, if you don't want combat strength, maximum weight, maximum moves, object required for search, the number of treasures needed before you have won or the number of monsters killed before you have won to take a part in the game, then enter 0 for each one.
- (5) The terms 'monsters' and 'treasures' should be interpreted widely, if you wish to get as much flexibility as possible from this program. Treasures can be anything that needs to be collected in the adventure, such as the pieces of a broken computer or the six missing people in a blazing building.
- (6) The 'magic' command, when used in the right location, transports the player to another room at a cost of a few moves and a couple of strength points.
- (7) Take note when you are entering the object names in, the computer will only accept the *full* object name. This means that if you enter the object name as 'A glittering, golden orb, the size of a human hand, encrusted with diamonds, sapphires, emeralds and many other jewels' then that is the name you would have to type in when playing the adventure, as in TAKE A glittering, golden orb etc.
- (8) The commands to play the adventure are pretty straightforward. They are listed between lines 4090 and 4130.

- (9) To win the adventure you must fulfil the three winning conditions of the number of monsters killed, the correct number of treasures, and being present in the correct winning location (held in the variable, WL).
- (10) You can lose in several ways. Firstly, you can run out of moves (when your MV value exceeds your MM value). If your combat strength is reduced to below zero, then you are finished and must start again. The third way to lose the game is to find yourself in a location with no way out. If you are in a location with only the magic command providing an exit, then the computer allows you a couple of goes at getting out, and if it is obvious that you do not know the magic command or you don't have your wits about you to realise that you are stuck in a place with no normal exits, then it finishes you off.
- (11) When locking and unlocking doors, the command is entered in the format 'UNLOCK N' or 'LOCK SOUTH'.
- (12) You may want to edit the data once you've played the game. Due to memory constraints (at one point I found I only had eight bytes left, hardly enough to run the cassette buffer!) there is no program editor. What you can do is break the program and then directly enter the changed item of data. From looking at the program, you can see that the main arrays and variables are:

LS(n) the location name array
L(x,y) the location map data
OB\$(n) the object name array
OB(x,y) the object characteristics
MOS(n) the monster name array
MO(x,y) the monster attribute data
R\$(n) the messages array
R(n) each messages position
N the number of locations
BM the number of monsters
OM the number of objects
P the original starting position for the player
MG\$(1) the answer given by the computer when the player asks for help.
MG\$(2) the magic command
MG\$(3) the title of the adventure

- (13) When you want to save the adventure on tape, press option 3 on the menu, type in the file name you wish the program to be saved as, and the computer will direct you from that point.

- 14) If you want to load the data back immediately after saving it, then the load option is chosen (option 4). Answer Y to the question, 'Have the useless arrays been erased?'. If you are loading a set of data that you have had on tape before this run, then answer N to the question. This ensures that the surplus arrays are erased, thus releasing enough memory for the cassette buffer to work as well as leaving enough memory for string handling.
- 15) As regards saving and loading files, I find I have somewhere around 90% success. Data files seem to be more susceptible to loading problems than normal programs. Always save a completed set of data on two different tapes. That way, if you have any problems with one tape (from little sister using it as a frisbee, to magnetic interference caused by leaving the cassette too near a T.V.) then you still have one in reserve.

Finally, I hope you have a lot of fun with this program and as a result find that you are able to write better adventures at the end of using it. Please feel free to fiddle around with the program, changing the available commands and the like to suit your own needs. I would be very interested to hear of any modifications or improvements that you make to my program or to receive copies of adventures written using *Pentacle*. Please write to me care of Interface Publications.

Amstrad PENTACLE

Castle of Doom

Here's the data for the *Castle of Doom* adventure.

We start with the fifteen location names depicting a castle and assorted 'nasty' places:

ON THE DRAWBRIDGE
BY THE PORTCULLIS
IN THE SLIME-FILLED MOAT
IN THE GREAT HALL
ON THE BATTLEMENTS
IN THE KITCHENS



IN A SECRET PASSAGEWAY
IN A DARK CAVE
STUCK IN THE CAVE SYSTEM
IN THE CASTLE DUNGEONS
IN A SNAKE-INFESTED PIT
IN THE TORTURE CHAMBER
IN THE LIBRARY
IN THE BANQUET ROOM
IN THE GREAT TOWER

From this we move to the map data for the adventure. Note that a minus sign signifies a locked door. You can either enter the data as it is, including the -13 and -10 and leave the section dealing with locked and unlocked doors, or you can type 0 for the two minus values and then enter the values in the locked/unlocked door section.

```
1 ) 0 3 3 2 3 0 0 0
2 ) 3 3 1 5 0 0 0 0
3 ) 0 0 0 0 0 0 0 0
4 ) 4 0 0 0 5 6 0 7
5 ) 0 14 0 12 0 4 0 3
6 ) 14 4 14 0 10 0 7 0
7 ) 0 6 8 9 0 0 0 0
8 ) 9 11 11 10 12 0 0 11
9 ) 0 0 0 0 0 0 7 0
10 ) -13 2 7 0 6 10 0 0
11 ) 0 0 0 0 0 0 0 0
12 ) 13 11 10 3 7 0 0 0
13 ) 4 14 -10 1 5 0 15 0
14 ) 6 0 0 7 0 0 0 8
15 ) 5 13 0 0 0 10 0 3
```

Here is the object name and attribute data:

RUSTY SWORD
BAG OF GOLD
SHINING AMULET
GLEAMING SAPPHIRE
RARE BOOK
STRING OF PEARLS
WOODEN CLUB



KEY

OAKEN SEAL

GOBLET STICKY WITH MEAD

| | | | |
|----|----|---|---|
| 2 | 12 | 0 | 2 |
| 8 | 15 | 2 | 3 |
| 9 | 2 | 0 | 3 |
| 12 | 1 | 0 | 3 |
| 13 | 7 | 0 | 3 |
| 14 | 4 | 2 | 3 |
| 12 | 25 | 0 | 2 |
| 4 | 3 | 2 | 4 |
| 7 | 8 | 0 | 0 |
| 6 | 9 | 1 | 0 |

The final data consists of the monster section, the message data (please excuse the puns) and the game rules (such as the magic command and the maximum number of moves).

AN OLD KNIGHT 2 17

A HIDEOUS OGRE 10 25

THE GUARDIAN 15 45

GO WEST YOUNG MAN 1

NORTH DOOR SHHH! 10

DESSERT MENU: BAT PIE AND I SCREAM 6

COMBAT STRENGTH 75

MAXIMUM CARRIED WEIGHT 30

HELP REPLY :- YOU ARE ON YOUR OWN

NUMBER OBJECT NEEDED FOR SEARCH 3

MAXIMUM NUMBER OF PLAYER MOVES 500

MAGIC COMMAND: PENTACLE

WINNING LOCATION 15

TREASURES NEEDED 5

MONSTERS SLAIN 3

PLAYER SEES EXITS

The Search for SPECK

Having graphics in adventures is a matter of taste. There is a large group of people who prefer their adventures to be text, and only text. They argue that graphics are often slow, undetailed and do not help develop the adventure's



plot. The large amounts of memory that they consume could be used for more text-described locations, puzzles and objects, they say.

I tend to agree with them, but I can't dismiss the impact of programs like *The Hobbit* and *Valhalla*, which utilised graphics to a great and, indeed, favourable extent. Much of the impact of the *Hobbit* stems from its graphics. Whatever your feeling about graphics, they are often found in adventure games.

This adventure is purely a graphical one, set in space. You must find, collect



and bring home a 'Socio-Political Economic Collection Capsule' that has strayed off its orbit. The capsule is of great importance as it is collecting data about the enemy planet's current life status (hence the title for the capsule).

You must search through the twenty-five sectors trying to avoid quasers, meteor storms and aliens. If your power holds out and you find the stricken capsule, then you must bring it back to ground safely. This is achieved by landing your vessel like a lunar lander, using thrust to slow your speed and reaching a height of less than 10 Kloms and an eventual velocity of less than eight parsecs.

Good luck Captain, keep your wits about you and conserve fuel where you can.

```
10 '      THE SEARCH FOR SPECK
20 '
30 GOSUB 1860
40 GOSUB 870
50 CLS
60 PAPER #1,3: PEN #1,2: PEN #2,1
70 '
80 '      MAIN LOOP
90 '
100 AA=A
110 WINDOW #2,6,35,22,25
120 WINDOW #1,5,36,21,25
130 INK 0,0: INK 1,0: BORDER 13: INK 2,9: IN
K 3,13
140 IF S(A)=3 OR S(A)=4 THEN 170
150 FOR T=1 TO 200: PLOT INT(RND*639)+1, I
NT(RND*300)+100, 1: NEXT
160 FOR T=200 TO 80 STEP-2: SOUND 1,T,1: N
EXT T
170 ON S(A) GOSUB 1030,1030,1160,1290,15
20,1620
```

```
180 PRINT #2,"SECTOR:";A;" POWER:";ST
190 PRINT #2,"SCAN:";P$(S(A))
200 PRINT #2," "
210 IF S(A)=5 THEN PRINT #2," PREPARE T
O LAND":WHILE INKEY$="":WEND:GOTO 470
220 INPUT #2," WARP TO SECTOR:";A
230 IF A>25 OR A<1 THEN 220
240 M=ABS(AA-A)
250 ST=ST-(M*100)
260 IF ST<1 THEN 380
270 '
280 ' WARP
290 '
300 MODE 0
310 FOR T=1 TO 100
320 PLOT 320,200
330 DRAW INT(RND*639)+1,INT(RND*399)+1,I
NT(RND*13)+1
340 NEXT
350 MODE 1
360 GOTO 80
370 '
380 ' LOSE
390 '
400 FOR T=1 TO 1500:NEXT
410 CLS:INK 1,11:PEN 1:INK 0,0,24
420 LOCATE 11,10:PRINT "YOU'RE ATOMISED!
!!"
430 SOUND 4,3000,500,15,1,1
440 FOR T=1 TO 4000:NEXT T
450 INK 0,0:END
460 '
```

```
470 ' THE LANDING
480 '
490 CLS
500 INK 2,9:INK 1,26
510 A=-35-INT(RND*60)
520 B=1600+INT(RND*380)
530 IF ST>4000 THEN C=(ST/10) ELSE C=250
+INT(RND*60)
540 CLS #1:CLS #2
550 PRINT #2,"HEIGHT:";B;"KLOMS"
560 PRINT #2,"VELOCITY:";A;"PARSECS"
570 PRINT #2,"FUEL:";C;"UNITS"
580 DRAWR -640,0,0
590 IF B<900 THEN Z=INT(400-(B/3)):PLOT
1,Z:DRAWR 640,0,2
600 INPUT #2,"ENTER THRUST";T
610 IF T<0 THEN T=0
620 IF T>C THEN T=C
630 C=C-T:B=B+A+(T-5)/2
640 A=A+((T-5)/2)-INT(RND*4):A=INT(A)
650 IF C<1 THEN CLS #2:PRINT #2," ":PRIN
T #2," OUT OF FUEL...":GOTO 380
660 IF (B<10 AND ABS(A)>8) OR B<-1 THEN
380
670 IF B<10 AND ABS(A)<8 THEN 700
680 GOTO 540
690 '
700 ' SUCCESS
710 '
720 CLS
730 LOCATE 12,12
740 PRINT "CONGRATULATIONS"
```

```
750 PRINT:PRINT
760 PRINT "YOUR COMMANDER IS MOST PLEASE
D WITH YOU"
770 PRINT:PRINT "    YOU HAVE BEEN GIVEN
  A PROMOTION"
780 FOR H=1 TO 2
790 RESTORE 840
800 FOR T=1 TO 25
810 READ G
820 SOUND 4,G*2,21,15,1
830 NEXT:NEXT:END
840 DATA 60,53,47,45,60,0,45,47,45,40,53
,0,53,47,45,36,40,40,45,45,47,53,47,60
850 END
860 '
870 '    INITIALISATION
880 '
890 A=1:RANDOMIZE TIME
900 ST=9000:MODE 1
910 SYMBOL AFTER 230
920 SYMBOL 231,198,118,126,27,252,126,63
,54
930 DIM S(25),P$(6)
940 RESTORE 960
950 FOR T=1 TO 25:READ S(T):NEXT
960 DATA 1,6,4,2,1,6,4,2,3,2,3,1,2,3,1,4
,6,2,3,2,2,4,3,1,4
970 FOR T=1 TO 6:READ P$(T):NEXT
980 DATA "Nothing but space debris","All
  clear","Meteor Storm"
990 DATA "QUASAR ahead","Stranded capsul
e","ALIENS!!!"
```

```
1000 J=INT(RND*8)+14:S(J)=5
1010 RETURN
1020 '
1030 '     GRAPHICS
1040 '
1050 '     CLEAR SPACE
1060 '
1070 INK 1,26
1080 PEN 3:FOR T=1 TO 3:LOCATE INT(RND*3
9)+1,INT(RND*20)+1:PRINT CHR$(238):NEXT
1090 PEN 2:LOCATE INT(RND*39)+1,INT(RND*
20)+1:PRINT CHR$(181)
1100 PEN 1:LOCATE INT(RND*39)+1,INT(RND*
20)+1:PRINT CHR$(176)
1110 PEN 2
1120 IF S(K)=5 THEN LOCATE 12,10:PRINT C
HR$(214);CHR$(215):LOCATE 12,11:PRINT CH
R$(213);CHR$(212)
1130 CLS #1:CLS #2
1140 RETURN
1150 '
1160 '
1170 '     METEOR STORM
1180 '
1190 ST=ST-500
1200 INK 2,7:PEN 2
1210 FOR T=1 TO 70
1220 LOCATE INT(RND*39)+1,INT(RND*19)+1
1230 PRINT CHR$(231)
1240 SOUND 1,INT(RND*1500)+400,15,15
1250 NEXT
1260 INK 1,26
1270 CLS #1:CLS #2
```

Amstrad PENTACLE

```
1280 RETURN
1290 '
1300 ' QUASARS
1310 '
1320 ST=ST-1000
1330 INK 1,26
1340 INK 2,24:PEN 2
1350 FOR V=150 TO 2 STEP -40

1360 FOR L=1 TO 360 STEP 4
1370 DEG
1380 PLOT 320+(V*1.5)*COS(L),249+V*SIN(L
),2
1390 NEXT: NEXT
1400 INK 2,24,0
1410 FOR R=25 TO 2 STEP -6
1420 FOR T=1 TO 360 STEP 6
1430 DEG
1440 PLOT 320+(R*1.5)*COS(T),249+R*SIN(T
),2
1450 NEXT: NEXT
1460 CLS #1:CLS #2
1470 FOR T=100 TO 250 STEP 1
1480 SOUND 1,T,2:SOUND 1,T+3,2:SOUND 1,T
-3,2
1490 NEXT
1500 INK 2,24
1510 RETURN
1520 '
1530 ' STRANDED CAPSULE
1540 '
1550 INK 1,26
```

```
1560 LOCATE 12,10:PRINT CHR$(214);CHR$(2
15):LOCATE 12,11:PRINT CHR$(213);CHR$(21
2)
1570 FOR T=200 TO 80 STEP -20
1580 SOUND 1,T
1590 NEXT
1600 CLS #1:CLS #2
1610 RETURN

1620 '
1630 ' ALIENS!
1640 '
1650 ST=ST-6000
1660 RESTORE 1770
1670 INK 2,7,11
1680 PEN 2
1690 CLS #1:CLS #2
1700 FOR X=1 TO 19:FOR Y=1 TO 19
1710 READ X,Y
1720 LOCATE X,Y
1730 PRINT ">@{"
1740 NEXT:NEXT
1750 LOCATE 19,10
1760 PEN 1:PRINT CHR$(154);CHR$(225);CHR
$(154)
1770 DATA 4,3,16,3,22,3,35,3,9,6,19,6,30
,6
1780 DATA 14,8,24,8,14,12,24,12,9,14,29,
14
1790 DATA 17,16,21,16,4,17,35,17,11,18,2
7,18,19,19
1800 INK 1,26
1810 ENT -1,20,4,1,20,-4,1
```

```
1820 SOUND 1,160,400,15,0,1
1830 FOR T=1 TO 3000:NEXT T
1840 RETURN
1850 '
1860 '      INSTRUCTIONS
1870 '
1880 CLS
1890 PRINT TAB(6);"THE SEARCH FOR S.P.E.
C.K"
1900 PRINT:PRINT:PRINT "  The vitally im
portant surveillance      of Andromeda Six
hinges on the efforts  of a special cap
sule collecting data    about the planet
's Socio-Economic and  Political status
"
1910 PRINT:PRINT "      Unfortunately, the
capsule has             strayed from its orbit
and has gone           missing. You must find
the capsule by        warping around the sec
tors and checking your scanner."
1920 PRINT:PRINT "      Once found, the ca
psule must be         carefully brought to g
round, so a           perfect landing is req
uired."
1930 PRINT:PRINT "      If you manage to res
cue the S.P.E.C.K. (Socio-Political-Econo
mic Collection        Kapsule), you will be
amply rewarded."
1940 PRINT:PRINT "                      G O O D
L U C K"
1950 WHILE INKEY$="":WEND:RETURN
```


Leaked!

A top secret report containing information that could seriously embarrass the government has gone missing. The P.M. has sent for you personally. You are a respected backbencher and have the time to try and find the report *drawing as little attention to yourself or your task as possible.*

You must get yourself a contact that may divulge the holders of the report. Exactly where from, no one knows, but what is certain is that if you don't find the report in the month allotted to you, then your chances of a place in the Cabinet are nil.

Wander around Whitehall, check for news at your private office, hear the debates in the Chamber or go to your appointed select committee. Whatever you do, you must recover the missing file before it is leaked to the media.

A knowledge of political institutions is needed in this adventure but not to any great degree and you will pick up much of the game simply by playing it a number of times. It's not a very common topic for adventure games but one which I think is most suitable. I was looking forward to writing an adventure with political characters in as well as institutions but I didn't want to ruin my chances of a place on the New Year's Honours List, so I came up with this program which dispenses with normal adventure input and instead offers a number of action choices which you select by pressing the appropriate numeric key.

```
10 REM*****  
20 REM***** 'LEAKED!' *****  
30 REM*****
```

```
40 GOSUB 560
50 REM****COMMAND LOOP****
60 MV=MV+1:IF MV>30 THEN PRINT "YOU TOOK
  TOO LONG...THE REPORT HAS BEEN LEAKED..
  .YOUR POLITICAL CAREER IS OVER.":END
70 GOSUB 140
80 GOSUB 400
90 IF P=15 THEN 180
100 GOSUB 310
110 IF P=-1 THEN P=5:PRINT:PRINT "The co
  rrespondent mentions that a memberof the
  Select Committee dealing with thematter
  may have information but points out th
  at all favours in politics are recipr
  ocal":WHILE INKEY$="":WEND
120 R(MV)=PP
130 GOTO 50
140 TX=0:FOR T=1 TO MV:IF R(T)=10 THEN T
  X=1
150 NEXT
160 IF TX=1 AND R(MV-1)=11 AND P=9 THEN
  P=15
170 RETURN
180 REM****REPORT HELD****
190 PRINT:PRINT "DO YOU 1: Give details
  to the serious press"
200 PRINT:PRINT " 2: Carefully exp
  lain the matter to the PLP"
210 PRINT:PRINT " 3: Meet Permanen
  t Secretary of dept. where leak occurr
  ed"
```

```
220 PRINT:PRINT "      4: Take the matt
er to the P.M. directly"
230 PRINT:PRINT "      5: Explain your
findings to the chairman of the 1922 co
mmittee"
240 PRINT:PRINT:PRINT "  ENTER YOUR DEC
ISION AND QUICKLY"
250 INPUT L
260 IF L=INT(109.564/49) THEN PRINT "WHA
T THE OPPOSITION? ARE YOU MAD?":P=16
270 IF L=INT(400/96) THEN P=17
280 IF L=(INT((900*PI)/450)-5) OR L=(371
70/12390) OR L=(20^-1)*100 THEN P=16
290 FOR T=1 TO 1300:NEXT T
300 GOTO 50
310 REM*****DECISION MAKING*****
320 REM*****AND RESULTS*****
330 PP=P
340 C$=INKEY$
350 IF C$="" THEN 340
360 IF C$<>"1" AND C$<>"2" THEN SOUND 1,
2000:GOTO 340
370 IF C$="1" THEN P=A(P)
380 IF C$="2" THEN P=B(P)
390 RETURN
400 REM*****SCREEN HANDLING*****
410 MODE 1:CLS:INK 1,0:INK 0,26:INK 2,6,
26:INK 3,11,26
420 PRINT:PRINT TAB(16);"LEAKED!"
430 SPEED INK 70,10
440 PRINT TAB(16);"====="
```


e rather extreme",1,"talk to him and accept an invitation to speak at a meeting tomorrow",3

710 REM

720 REM 3

730 DATA "You're at a meeting of the pressure group","make your speech and leave",8,"look around for some help",4

740 REM

750 REM 4

760 DATA "That course of action yielded no help at all.What's your next move","go back to the Chamber",6,"go to your private office",8

770 REM

780 REM 5

790 DATA "In the Select committee, a member seems desperate to press for clause 17 to be deleted, the majority of the committee seem against him"

800 DATA "side with the majority, clause 17 was rather controversial",6,"side with the individual M.P. and struggle to get the clause through",10

810 REM

820 REM 6

830 DATA "You are seated in the chamber, a debate is taking place on a controversial issue","stay and listen",7,"discuss the matter with the minister",4

840 REM

```
850 REM                                7
860 DATA "The debate has ended, voting h
as begun", "vote aye",13,"vote nay",12
870 REM
880 REM                                8
890 DATA "You are at your office, severa
l invitations are lying in your 'IN' tra
y", "Accept the offer to appear on a lead
ing current affairs program to boost you
and your party's flagging image",14
900 DATA "go to the cocktail party hoste
d by your party's national committee",11
910 REM
920 REM                                9
930 DATA "You are wandering through them
aze of Whitehall corridors", "explain
to a passing civil servant your reasons
for being there and ask the way to your
contact's office",16,"Give up, you'll ne
ver find his office",8
940 REM
950 REM                                10
960 DATA "The grateful M.P. gives you de
tails of the one person who can help", "
return to the lobby",1,"visit Whitehall"
,9
970 REM
980 REM                                11
990 DATA "At the cocktail party there ar
e a large number of honoured guests incl
uding many leading ministers, the P.M. an
```

d a couple of Permanent Secretaries.

1000 DATA "discuss with them, how far you have got being careful to check that no one over-hears",16

1010 DATA "Have a chat with one of the Permanent Secretary's assistants over the strange and often confusing layout of Whitehall",9

1020 REM

1030 REM 12

1040 DATA "Your surprising vote against the party line in yesterday's major debate has caused some interest. One of the most influential lobby correspondents has come to talk to you"

1050 DATA "explain that you felt it was an issue of conscience and that you are solidly behind all other party policies",13

1060 DATA "explain in the strictest confidence the situation the government is in and does he know of any 'sources' of information",-1

1070 REM

1080 REM 13

1090 DATA "While dining with three Shadow ministers in St.Stevens tavern, the subject of leakages comes up","continue and mention your problem",16,"State that you've never had to deal with such a problem",1


```
1100 REM
1110 REM                14
1120 DATA "You are on the chat show and
are asked the question, 'How are your r
esearches into the missing Heathcliffe
report going?'"
1130 DATA "reply, 'no comment'",16,"repl
y, 'I'm sorry Robin, I didn't know that
any reports were missing. I think you
should check your sources'",4
1140 REM
1150 REM                15
1160 DATA "You now have the leaked repor
t and the source of the leakage",0,0,0,
0
1170 REM
1180 REM                16
1190 DATA "You have brought too much att
ention to yourself and your activites.
The P.M. isnot pleased with you...your c
hances of apromotion are hignly unlikely
",0,0,0,0
1200 REM
1210 REM                17
1220 DATA "You have made the right decis
ions and your handling of this whole a
ffair has brought you much credit withi
n the party. You are given a place
in the Cabinet...your political futu
re seems tobe assured.",0,0,0,0
```

Creating Your Own

Writing adventures is as much fun, if not more, than *playing* adventures. In addition to the fun derived, it is a very good way of improving your programming. It is not as difficult as you may think, and in these pages I will try to give a few helpful hints on writing a good adventure game.

There are two basic sides to writing an adventure, and both must be planned for. The plot, the setting of the adventure, the type of creatures, objects and puzzles and the commands at your disposal all add up to what can be conveniently called the *story board*. The second part is the actual programming of the adventure, to put the story board into a running program.

Let's deal with the story board first. It is best to work out your plot first. This really consists of deciding what must the player do to win the adventure along with the setting and location for the adventure. In these areas, you should use your imagination as much as possible.

There are too many adventures based on the familiar concepts of fighting medieval monsters and looking for a missing princess. If you want to write a good adventure today, then you must add some novel elements to the plot.

Computer Adventure

As a humble debugging program, you must search around for the bugs that exist in the computer's operating system.

The Day After the Night Before

You had rather a good party at your house last night, too good as you hold your aching head. Your parents are coming home today and the place is in a shambles. Can you get your home looking spick and span and so avoid incurring your parents' wrath?

The Ten Tasks of Hercules

You must try to emulate Hercules by taking on ten extremely arduous and near-impossible tasks. Your only problem is what tasks. You must keep yourself in prime condition while always being on the look-out for another of

the ten tasks to complete. You've got ten weeks before the gods check your progress.

These adventure ideas are all a little more novel than the classic styles of undersea treasure exploration or blood and thunder monster battling.

Puzzles

Moving on from the plot, one must determine the numbers and the nature of the puzzles and the objects and commands needed to solve them. More than any other thing, the puzzles make the adventure. You must decide upon the complexity of the adventure and construct your puzzles accordingly.

One way of writing puzzles that fit in well with your adventure plot is to split the puzzles and situations into two distinct types. There is one type which must be solved for the adventure to be completed, while the second group are additional puzzles adding to the realism of the adventure or its complexity. The player is not told which puzzles fit into which categories, making his or her task harder. You can stick carefully to your plot for the major puzzles, and then include as many of the incidental problems as memory or time allow.

Obviously, the commands needed and the objects available will be largely determined by the puzzles in your adventure. As with the puzzles, it is worth having some definite 'red herrings' which can trick the player. Below is a small example story board to show you how the principles already mentioned can be put into practice.


Underground Plague

Setting:

The London Underground and selected London tourist spots.

Plot:

Giant rats are breeding in the underground system, they are carriers of the



bubonic plague and are slowly gathering in numbers before they come to the surface to attack London.

Locations:

Easily derived from the plot; take a look at a London underground map and throw in famous tourist spots such as Marble Arch, Buckingham Palace and the Tower of London.

Puzzles that need to be solved:

Find the rats' hiding place. Find something to eliminate them with, and do it in time.

Puzzles that add realism etc:

Darkness in the underground tunnels; oncoming tube trains; electric lines.

Objects:

Torch; wirecutters; insulated gloves; vaccine; rat-gun (a red herring); flamethrower (the real weapon).

Programming

Once you have got your story board, you have to put it into program form. The exact way in which you do this varies with each programmer's preference as much as the story board varies but there are some essential points that are worth keeping in mind. Grammar and proper spelling are essential. It is foolish to spend a lot of time writing an adventure only to find that you have left spelling mistakes in. Always check out spelling with a dictionary if you are not sure.

Presentation is also important. Before someone really gets involved in an adventure, their first impression is largely derived from the screen layout and tidiness. Avoid cutting words in two at the end of a screen line. If you have spare memory (and unless your adventure is huge, the Amstrad's ample 43K should leave plenty of it spare) add colour and sound effects just to liven things up here and there.

Remember that the game should be fun to play, be quite friendly with the comments that the computer makes to the player. A sense of humour is appreciated as well.

Perhaps the most important programming factor that springs to mind is that of 'structuring'. Program structuring with adventure games is essential, it

saves memory (very important if you are writing a large adventure), improves the efficiency and hence the speed of the adventure and also makes debugging an easier task. If you have structured an adventure well, then you should be able to return to it several months later and immediately know where each section of code is, and what each routine does.

To structure your program, it is well worth extensively using **REM** statements. Looking at my programs, you can see that I've tried to use variable names that go with their task, such as **OB** for objects and **L** for location data.

With these points in mind, I hope you can now think more seriously about writing your own adventure game of quality.

50 Dynamic Games for the Amstrad!

Here's a magnificent collection of fast-moving games from Clive Gifford, author of the PENTACLE. You'll find arcade action, stimulating board games, more adventures, challenging adventures, and several programs just for fun. Whatever kind of games you want, you'll get them in this collection.

Here's what is in store for you in this £5.95 book:

CHESSE REVERSI GOMUKU FOUR-IN-A-ROW
CHECKERS SYNTHETIC STAR THEME ZAP ENERGISER
CHOPPER RUSTLE, RUSTLE MUSIC COMPOSER
MUSIC MAZE CUBE-MASTER SOLITAIRE FLIPPA
ECHO CHAMBER SCRAMBLER REPEAT PONTOON
DOUBLE DICE BIRD CAGE GRAPHIC CARDS
UNDER OR OVER ONE-ARMED BANDIT
CITY OF ATLANTIS EXPLORER CHARACTER GENERATOR
STEINBERG FILES SNIPER MATCHSTICK MEN
PITCH 'N' PUTT NEEDLE IN HAYSTACK ROM BUG
TRIANGULAR TRACER WINDOW IN/OUT LASER FAN
LUNATIC PATCHWORKS HUNDREDS AND THOUSANDS
SUNBURST TRAVELLING TRIANGLES DANGER UXB
QUACK ATTACK DIAMOND THIEF BOMB CATCH
BRICK OUT CITY SACRIFICE ATTACK OF THE MUTOIDS
SURROUND ARNOLD THE ELECTRICIAN

Buy it where you bought THE AMSTRAD PENTACLE or, in case of difficulty, send us the coupon below. We'll pay the postage.

Interface Publications,
Dept AP,
9-11 Kensington High Street,
London W8 5NP



Please send me a copy of DYNAMIC GAMES FOR THE AMSTRAD by Clive Gifford. I enclose £5.95.

Name _____

Address _____

Amstrad PENTACLE

It's suddenly very easy to create your own adventures on the Amstrad. Just type in the PENTACLE program from this booklet, and then follow the simple instructions to create as many exciting, hard-to-solve adventures as you like.

You simply enter the data for your own adventures, and the PENTACLE does **all** the hard work, creating an adventure you can save, and even sell (without paying us a penny in royalties!).

As a bonus, we've also included a full set of data to show PENTACLE in use with an adventure set in bygone days, **Castle of Doom**. A change of pace is provided with the full graphical adventure, **The Search for SPECK**.

Adventuring with your Amstrad will never be the same again!



£2.99

ISBN 0-907563-70-8



9 780907 563709